Numerický výpočet integrálu

V mnoha případech nemůže *Mathematica* dát přesnou hodnotu integrálu a pak je nutné počítat přibližnou hodnotu. Toto téma spadá spíše do numerické analýzy a tak uvedeme jen stručně několik příkladů.

```
Integrate[x^x, {x, 0, 1}]
```

$$\int_0^1 x^x \, dx$$

```
NIntegrate[x^x, {x, 0, 1}]
```

0.783431

Mathematica zná mnoho numerických metod k výpočtu integrlů.Je možné ji v požadavcích specifikovat. My se omezíme na výpočet pomocí Riemannových součtů. Lze brát body v pravých nebo levých krajních bodech dělících intervalů. Už na 5.desetinném místě je vidět rozdíl v hodnotách.

```
NIntegrate[x^x, {x, 0, 1},
 Method → {"RiemannRule", "Type" → "Right"}]
```

— *NIntegrate::slwcon :*

Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. ≫

— *NIntegrate::ncvb :*

NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x near {x} = {0.0644531}. NIntegrate obtained 0.783437 and 0.0000259724 for the integral and error estimates. ≫

0.783437

```
NIntegrate[x^x, {x, 0, 1},
 Method → {"RiemannRule", "Type" → "Left"}]
```

0.783424

Riemannův integrál

Riemannova definice integrálu jako limita jistých součtů je vlastně metoda přibližného výpočtu integrálů. Nehodí se však k tomuto účelu, protože konverguje velmi pomalu. Ukážeme nyní na příkladě, jak dolní a horní Riemannovy součty k integrálu konvergují a jak se tyto součty chovají k ploše mezi grafem a osou x. Za funkci zvolíme polynom 4.stupně a budeme kreslit příslušné obdélníky pro dělení intervalu na 10 až 30 stejných intervalů. Nejdříve dolní součty, jejichž hodnota je v grafu uvedena vpravo pod hodnotou integrálu.

```
Clear[f]; f[x_] := x^4 - 3 x^2 + 1;
Plot[f[x], {x, -1, 2}]
```

```
Clear[fmin];
fmin[n_] :=
 Table[MinValue[{f[y], -1 + 3 / n * k ≤ y <= -1 + 3 / n * (k + 1)},
    y] // N, {k, 0, n - 1, 1}]
Clear[smin];
smin = Table[Sum[3 * fmin[n][[k]] / n, {k, 1, n, 1}],
   {n, 10, 30, 1}];
Clear[tmin];
tmin =
 Table[
  Table[Rectangle[{-1 + k * 3 / n, 0},
    {-1 + (k + 1) * 3 / n, fmin[n][[k + 1]]}], {k, 0, n - 1, 1}],
   {n, 10, 30, 1}];
```

```
Manipulate[Plot[f[x], {x, -1, 2},
  Prolog -> {EdgeForm[Thick], Pink, tmin[[n - 9]],
    Text[Style["∫f[x]ⅆx=0.6", Black], {2.5, 4}],
    Text[Style["dolní součet", Black], {2.5, 3}],
    Text[Style[smin[[n - 9]], Black], {2.5, 2.6}]},
  Axes → True, AxesOrigin → {0, 0},
  PlotRange → {{-1, 3}, {-1.4, 5}}], {n, 10, 30, 1}]
```



Nyní uděláme totéž pro horní součty.

```
Clear[fmax];
fmax[n_] :=
 Table[MaxValue[{f[y], -1 + 3 / n * k ≤ y <= -1 + 3 / n * (k + 1)},
    y] // N, {k, 0, n - 1, 1}]
Clear[smax];
smax = Table[Sum[fmax[n][[k]] * 3 / n, {k, 1, n, 1}],
   {n, 10, 30, 1}];
Clear[tmax];
tmax =
 Table[
  Table[Rectangle[{-1 + k * 3 / n, 0},
     {-1 + (k + 1) * 3 / n, fmax[n][[k + 1]]}], {k, 0, n - 1, 1}],
  {n, 10, 30, 1}];
```

```
Manipulate[Plot[f[x], {x, -1, 2},
  Prolog -> {EdgeForm[Thick], Pink, tmax[[n - 9]],
    Text[Style["∫f[x]ⅆx=0.6", Black], {2.5, 4}],
    Text[Style["horní součet", Black], {2.5, 3}],
    Text[Style[smin[[n - 9]], Black], {2.5, 2.6}]},
  Axes → True, AxesOrigin → {0, 0},
  PlotRange → {{-1, 3}, {-1.4, 5}}], {n, 10, 30, 1}]
```



Rosdíl mezi horními a dolními součty je znázorněn na dalším grafu, opět s příslušnými číselnými hodnotami.Jako poslední položka je uveden velký rozdíl mezi hodnotou integrálu a Riemannovými součty pro naši funkci.

```
Clear[tsub];
tsub =
 Table[Table[Rectangle[{-1 + k * 3 / n, fmin[n][[k + 1]]},
    {-1 + (k + 1) * 3 / n, fmax[n][[k + 1]]}], {k, 0, n - 1, 1}],
  {n, 10, 30, 1}];
```

```
Manipulate[Plot[f[x], {x, -1, 2},
  Prolog -> {EdgeForm[Thick], Pink, tsub[[n - 9]],
    Text[Style["rozdíl součtů", Black], {2.5, 3.5}],
    Text[Style[ smax[[n - 9]] - smin[[n - 9]], Black],
     {2.5, 3}]}, Axes → True, AxesOrigin → {0, 0},
  PlotRange → {{-1, 3}, {-1.4, 5}}], {n, 10, 30, 1}]
```



```
ColumnForm[{Integrate[f[x], {x, -1, 2.}] "=∫f(x)ⅆx",
  smin[[21]]
   "=dolní Riemannův součet pro dělení na 30 intervalků",
  smax[[21]]
   "=horní Riemannův součet pro dělení na 30 intervalků"}]
```

0.6 =∫f(x)ⅆx
0.08999 =dolní Riemannův součet pro dělení na 30 intervalků
1.13963 =horní Riemannův součet pro dělení na 30 intervalků